# Maintainer: Levente Polyak <anthraxx[at]archlinux[dot]org>
# Contributor: Daniel Micay <danielmicay@gmail.com>
# Contributor: M Rawash <mrawash@gmail.com>
# Contributor: János Illés <ijanos@gmail.com>

pkgname=vim-fugitive
pkgver=3.7
pkgrel=1
pkgdesc='Git wrapper so awesome, it should be illegal'
url='https://github.com/tpope/vim-fugitive'
arch=('any')
license=('custom:vim')
depends=('vim' 'git')
groups=('vim-plugins')
source=("${url}/archive/v${pkgver}/${pkgname}-${pkgver}.tar.gz"
        license.txt)
sha512sums=('8252b69c6183f5b710739eb68ecb3fbdfad52df3dbee0a3a04ad33f410b489d7fcf1c0bde0cb2d53995a6b4e34191c165c652d08e6145789a88f8c46bfcddf56'
            'a50e91b1896b0d952008ba2f641a87af2d1a01e4f280f6c914edcd51ae5d1586d5ade71c3609866b501569007bcb7f2494f08280afec170884b90fab36332fac')
b2sums=('1c00272174bd76a2b4d34aa54a7498b2f75a5322d587e3436d82f21d6188027768bea0475c99fcbc50a82b89d9a9a7697d18106cb1eb2f9788afa81848fe2de8'
        '63a85fc6710dc62cf9d982eaf8fa048ccc81754e9c67c6a071ae9608c7ba90f07d744733f377e08078612dcc9a0e33590c96f4a4a1f81cdba72f86bddf34e324')

package() {
  cd ${pkgname}-${pkgver}
  local _installpath="${pkgdir}/usr/share/vim/vimfiles"
  install -d "${_installpath}"
  cp -r -t "${_installpath}" autoload doc plugin ftdetect ftplugin syntax
  install -Dm 644 ../license.txt -t "${pkgdir}/usr/share/licenses/${pkgname}"
}

# vim: ts=2 sw=2 et:
